
 

Name: AH         Cat HW SW BASIC   Page 1 of 3 

1. Applicability 
This application note is applicable to systems using Drives and Motors with a Trio Motion Coordinator, 
where the Motion Coordinator is controlling the position servo loop. 

2. Introduction 
On some applications, it is necessary to disable the drives when a machine guard or door is opened, 
but then to maintain the original axis positions after the drives are enabled again.   

Trio recommend that the operation of safety guards is controlled by a recognised external circuit 
such as a certified safety relay.  Part of such a safety circuit might include a connection from the 
Motion Coordinator’s Watchdog Relay output. 

TO REDUCE THE POSSIBILITY OF UNCONTROLLED MOTION, THE MOTION COORDINATOR AXES’ DEMAND 
POSITIONS WILL AUTOMATICALLY ADJUST THEMSELVES TO FOLLOW THE FEEDBACK POSITIONS WHILE 
WDOG IS OFF.(1) 

This feature(2) prevents the axes from “jumping” back to some previous set demand position when the 
WDOG=ON command is given.  However in some applications, it is necessary to restore the original 
axes’ positions after the WDOG=ON command and this should be done under program control. 

 
1 Applies to system software version 1.6444 and greater. 
2 Any operational feature that depends on software cannot be considered 100% safe.  Means of protecting people operating the 
machine must exist external to the Motion Coordinator. 

3. Restoring the axes to their required positions 
When the WDOG=OFF command is used DPOS tracks MPOS; so if the axes, which are disabled, are 
moved by hand; they will be out of position when the WDOG=ON command is given and they will not 
automatically move back to their original positions.  If the demand position is stored before the 
watchdog is turned off, then the original positions can be regained later.   

This example program routine will store the axes’ DPOS values prior to issuing the WDOG=OFF 
command.  It then restores the axes’ to their previous positions by use of a controlled move after the 
WDOG=ON.  The example is based on a 4 axis application. 

' toggle watchdog, but maintain position 
 
' record the DPOS values before turning off the watchdog 
VR(0)=DPOS AXIS(0) 
VR(1)=DPOS AXIS(1) 
VR(2)=DPOS AXIS(2) 
VR(3)=DPOS AXIS(3) 
 
WDOG=OFF 
 
WDOG=ON 
 
' move the axes under control to their correct positions 
' (assumes SPEED, ACCEL and DECEL are set for each axis) 
MOVEABS(VR(0)) AXIS(0) 
MOVEABS(VR(1)) AXIS(1) 
MOVEABS(VR(2)) AXIS(2) 
MOVEABS(VR(3)) AXIS(3) 
WAIT IDLE AXIS(0) 
WAIT IDLE AXIS(1) 
WAIT IDLE AXIS(2) 
WAIT IDLE AXIS(3) 

 

Where an axis moves across the REP_DIST in a continuous rotating machine, an IF statement will be 
needed in the software to decide which way to move back to the stored position.  So for example if 
VR(0) is a value near the end of the REP_DIST and DPOS is near to 0 after the watchdog is turned back 
on, the axis will move back not forward one revolution. 

BASE(1) 
d=DPOS-VR(1) ' calculate axis displacement 
n=REP_DIST/2 ' calculate half rep_dist 
PRINT DPOS,VR(1),d,n 
IF d >= n THEN 
  MOVEABS(VR(1) + REP_DIST) 
ELSEIF (d < n) AND (d > -n) THEN 
  MOVEABS(VR(1)) 
ELSEIF (d < -n) THEN 
  MOVEABS(VR(1) - REP_DIST) 
ENDIF 

 

The above routine is a “shortest path” calculation.  For some machines this might not be the correct 
response; for example when an axis must only move forward. 

If necessary a test should be done to decide if the axes have moved too far while the watchdog was 
off; and therefore an error condition will exist and the machine cannot continue without some 
intervention from an operator. 

4. Disabling the servo-drives without turning off the watchdog 
If a system designer wishes to disable the drives without turning off the Motion Coordinator’s 
watchdog relay, then this is possible. 

HOWEVER TRIO MOTION TECHNOLOGY DOES NOT RECOMMEND THIS APPROACH. 

If the axes are moved out of position while the drives are disabled and the WDOG=ON, then upon re-
enabling the drives, the axes may move rapidly out of control back to their original positions.  This 
might cause damage or personal injury. 

SYSTEM DESIGNERS MUST MAKE A FULL RISK ASSESSMENT BEFORE USING THE CIRCUIT SHOWN HERE. 

The circuit shown below can be used to disable the servo drives under program control without 
turning off the watchdog relay.  It also allows the following error limit detection (FE > FE_LIMIT) to 
continue to protect the axes by wiring the controlled output and watchdog relay in series. 

 

 
Fig 1.  Digital output passed through watchdog relay 

 

Where a program needs to disable the drives, the OP command is used: 

' enable the drives on power-up 
OP(6,ON) 
WDOG=ON 
 
' Disable drives 
OP(6,OFF) 
 
' Re-enable drives 
OP(6,ON) ' WARNING motors may jump! 

 
When any axis has a MOTION ERROR detected by the Motion Coordinator, the watchdog relay opens 
and the drives are disabled. 

